Worksheet: **Lecture1.mws**

Lecture 1: What is a PDE?

Andrew J. Bernoff

PCMI, Summer 2003

This worksheet contains the examples from the first lecture.

Exercise 1: Verifying Some Solutions to Laplace's Equation

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
```

```
> U:=x;
```

$$U := x$$

```
> Uxx:=diff(U,x$2);
```

$$Uxx := 0$$

```
> Uyy:=diff(U,y$2);
```

$$Uyy := 0$$

```
> Uxx+Uyy;
```

$$0$$

```
> plot3d(U,x=-1..1,y=-1..1,style=patchnogrid,shading=ZHUE,axes=boxed);
```



```
> V:=x^2-y^2;
```

$$V := x^2 - y^2$$

```
> Vxx:=diff(V,x$2);
```

$$Vxx := 2$$

```
> Vyy:=diff(V,y$2);
```

$$Vyy := -2$$

```
> Vxx+Vyy;
```

$$0$$

```
> plot3d(V,x=-1..1,y=-1..1,style=patchnogrid,shading=ZHUE,axes=box
  ed);
```




```
> W:=a*U+b*V;
```

$$W := a\,x + b\,(x^2 - y^2)$$

```
> Wxx:=diff(W,x$2);
```

$$Wxx := 2\,b$$

```
> Wyy:=diff(W,y$2);
```

$$Wyy := -2\,b$$

```
> Wxx+Wyy;
```

$$0$$

```
> Wplot:=c*U+(1-c)*V:
> animate3d(Wplot,x=-1..1,y=-1..1,c=0..1,style=patchnogrid,shading
  =ZHUE,axes=boxed,frames=50);
```

Exercise 2: Verifying Solutions to the Minimal Surface Equation

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
> Z:=log(sin(y)/sin(x));
```

$$Z := \ln\left(\frac{\sin(y)}{\sin(x)}\right)$$

```
> PDE:=(1+Zy^2)*Zxx+2*Zx*Zy*Zxy+(1+Zx^2)*Zyy;
```

$$PDE := \left(1+\frac{\cos(y)^2}{\sin(y)^2}\right)\left(\frac{\cos(x)^2}{\sin(x)^2}+1\right)+\left(\frac{\cos(x)^2}{\sin(x)^2}+1\right)\left(-1-\frac{\cos(y)^2}{\sin(y)^2}\right)$$

```
> Zx:=diff(Z,x);Zy:=diff(Z,y);Zxx:=diff(Zx,x);Zyy:=diff(Zy,y);Zxy:
  =diff(Zx,y);
```

$$Zx := -\frac{\cos(x)}{\sin(x)}$$

$$Zy := \frac{\cos(y)}{\sin(y)}$$

$$Zxx := \frac{\cos(x)^2}{\sin(x)^2}+1$$

$$Zyy := -1-\frac{\cos(y)^2}{\sin(y)^2}$$

$$Zxy := 0$$

```
> PDE;simplify(PDE);
```

$$\left(1+\frac{\cos(y)^2}{\sin(y)^2}\right)\left(\frac{\cos(x)^2}{\sin(x)^2}+1\right)+\left(\frac{\cos(x)^2}{\sin(x)^2}+1\right)\left(-1-\frac{\cos(y)^2}{\sin(y)^2}\right)$$

$$0$$

```
> plot3d(Z,x=0..Pi,y=0..Pi,style=patchcontour,shading=zhue,axes=bo
  xed,grid=[100,100],title="Scherk's Minimal
  Surface",view=-1..1);
```



```
> X:=r*cos(phi)-r^3*cos(3*phi)/3;Y:=r*sin(phi)+r^3*sin(3*phi)/3;Z:
  =r^2*cos(2*phi);
```

$$X := r\cos(\phi) - \frac{1}{3}r^3\cos(3\,\phi)$$

$$Y := r\sin(\phi) + \frac{1}{3}r^3\sin(3\,\phi)$$

$$Z := r^2\cos(2\,\phi)$$

```
> plot3d([X,Y,Z],r=0..3,phi=-Pi..Pi,style=patchnogrid,color=phi,ax
  es=boxed,grid=[100,100],title="Enneper's Minimal Surface");
```



A Solution to the Convection Equation

```
> restart:
> with(plots):
```

Warning, the name changecoords has been redefined

```
> F:=x->exp(-x^2);
```

$$F := x \to e^{(-x^2)}$$

```
> xi:=x-C*t;
```

$$\xi := x - C\,t$$

```
> U:=F(xi);
```

$$U := e^{(-(x-C\,t)^2)}$$

```
> Ut:=diff(U,t);
```

$$Ut := 2\,(x - C\,t)\,C\,e^{(-(x-C\,t)^2)}$$

```
> Ux:=diff(U,x);
```

$$Ux := (-2\,x + 2\,C\,t)\,e^{(-(x-C\,t)^2)}$$

```
> PDE:=Ut+C*Ux;
```

$$PDE := 2\,(x - C\,t)\,C\,e^{(-(x-C\,t)^2)} + C\,(-2\,x + 2\,C\,t)\,e^{(-(x-C\,t)^2)}$$

```
> simplify(PDE);
```

$$0$$

```
> C:=1;animate(U,x=-10..10,t=0..10,numpoints=200,color=blue,thickn
  ess=2,title="Solution to the Transport Equation");
```

$$C := 1$$